STAT505 Assessment #7

In this study, eight people were given drug A, and another eight people were given drug B. Each person's heart rate was measured at four time points (after 5, 10, 15, and 20 minutes) from when the drug was given. The data are available in the "drugs.dat". Columns correspond to person, heart rate measurements one through four, and drug type.

(a) Using the split-plot ANOVA model, test for interaction between time and drug. Report the degrees of freedom and p-value. Comment on the evidence for interaction.

The null hypothesis is H_0 : no interaction between time and drug. The degrees of freedom for the hypothesis and error are 3 and 42, respectively. The p-value is less than .0001, which is significant evidence to reject H_0 and say that the effect of drug is not the same for all time points.

(b) Using the same model as above, test for equal drug effects. Report the degrees of freedom and p-value. Provide an interpretation of these results in relation to your results in part (a).

The null hypothesis is H_0 : no drug effect. The degrees of freedom for the hypothesis and error are 1 and 14, respectively. The p-value is .0219, which is significant evidence to reject H_0 and say that there is a drug effect. However, this effect is estimated by averaging over time points, which do not have equal drug effects as indicated by the significant interaction results in part (a). So, it is not obvious whether one drug is "better" or "worse" overall.

(c) Fit the MANOVA model to this data, and create histograms of the residuals for each of the four heart rate measurements (you don't have to include them here). Comment on the assumption of normality.

The histograms for the first two time points appear slightly skewed left, but the histograms for the last two time points appear symmetric and could arguably have come from normal distributions.

(d) Conduct a test of equal covariance matrices for the two drug groups. Report the p-value. Comment on the assumption of equal covariance matrices.

The null hypothesis is H_0 : equal covariance matrices for the two drug groups. The p-value is .7733, which is not significant evidence to reject the assumption of H_0. So, equal matrices are assumed, and the sample matrices will be pooled together.

(e) Provide a profile plot for the data with time along the horizontal axis. Comment on the evidence for interaction between time and drug type. Also comment on the evidence for a drug effect on heart rate.

The plot below shows non-parallel lines, which means the mean heart rates for each drug are not equal for all times. This suggests interaction between drug and time. The drug (main) effect is not as clear to define because it depends on which time point is considered. For time points 5, 10, and 20, the mean heart rates are higher for drug B. However, at time point 15 drug A is higher.

(f) Using the MANOVA model, conduct a test for interaction between time and drug. Report the test statistic, degrees of freedom, and p-value. What is your conclusion at the .05 level of significance?

The null hypothesis is H_0 : no interaction between time and drug. The Wilk's Lambda test statistic is 34.65 with 3 and 12 degrees of freedom. The p-value is less than .0001, which is significant evidence that drug and time do interact.

(g) Using the MANOVA model, conduct a test for equal drug effects by averaging over time. Report the test statistic, degrees of freedom, and p-value. What is your conclusion at the .05 level of significance? Provide an interpretation of these results in relation to your results in part (f).

The null hypothesis is H_0 : no drug effect. The Wilk's Lambda test statistic is 6.65 with 1 and 14 degrees of freedom. The p-value is .0219, which is significant evidence to reject H_0 and say that there is a drug effect. As before, this effect is estimated by averaging over time points, which do not have equal drug effects as indicated by the significant interaction results in part (f). So, it is not obvious whether one drug is "better" or "worse" overall.

(h) Comment briefly how the test in part (g) differs from the overall MANOVA test considered in the lessons last week.

The test above in part (g) is comparing the treatment (drug) groups by averaging over the components (time points), whereas the MANOVA test from last week compared treatment groups at each component separately.

(i) Note that the tests in parts (a) and (f) have the same null hypothesis, but the models provide different amounts of evidence to reject it. Briefly explain how the approach for part (a) is different from that for part (f).

Both the split-plot model and the MANOVA models assume that the various time responses for the same person are correlated, but the split-plot model further assumes that this correlation is equal for any two time points; the MANOVA model does not assume this.



SAS code:

```
data drugs;
  infile 'v:\drugs.dat';
  input person t1-t4 drug $ @@;  run;
data drugs2;
  set drugs;
  time=5;  rate=t1; output;
  time=10;  rate=t2; output;
  time=15;  rate=t3; output;
  time=20; rate=t4; output;
  drop t1 t2 t3 t4;  run;
proc glm data=drugs2;
  class drug person time;
  model rate=drug person(drug) time drug*time;
  test h=drug e=person(drug);
  run; quit;
proc glm data=drugs;
  class drug;
  model t1-t4 = drug;
  output out=resids r=rt1-rt4;
  run; quit;
proc univariate data=resids;
  histogram rt1-rt4;
  run;
proc discrim data=resids pool=test;
  class drug;
  var rt1-rt4;  run;
proc sort data=drugs2;
  by drug time;  run;
proc means data=drugs2;
  by drug time;
  var rate;
  output out=b mean=mean;  run;
proc gplot data=b;
  axis1 length=4 in;
  axis2 length=6 in;
  plot mean*time=drug / vaxis=axis1 haxis=axis2;
  symbol1 v=J f=special h=2 i=join color=black;
  symbol2 v=K f=special h=2 i=join color=black;
  symbol3 v=L f=special h=2 i=join color=black;
  symbol4 v=M f=special h=2 i=join color=black;
  run;
proc glm data=drugs;
  class drug;
  model t1 t2 t3 t4=drug;
  manova h=drug m=t2-t1,t3-t2,t4-t3;
  manova h=drug m=t1+t2+t3+t4;
  run; quit;
```